Maple-assisted proof of formula for A200573

Robert Israel

28 June 2018

There are $3^4 = 81$ configurations for a 2×2 sub-array. Consider the 81×81 transition matrix T with entries $T_{ij} = 1$ if the bottom two rows of a 3×2 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the top row of configuration i is the bottom row of configuration j, and no average of an element of the middle row and its neighbours is 1), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 4-element lists, corresponding to base-3 digits, in the order

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}.$$

```
> Configs:=  [seq(convert(3^4+i,base,3)[1..4],i=0..3^4-1)]:
> Compatible:= proc(i,j)
   if not Configs[i][1..2]=Configs[j][3..4] then return 0 fi;
   if Configs[j][1]+Configs[i][1]+Configs[i][2]+Configs[i][3] <> 4
  and Configs[j][2]+Configs[i][1]+Configs[i][2]+Configs[i][4] <> 4
   then 1 else 0
   fi
  end proc:
> T:= Matrix(81,81,Compatible):
```

Thus for $n \geq 2$ $\;a(n) = u\, T^{n-2} v$ where u is a row vector with entries 1 for configurations where the average of no bottom row entry and its neighbours is 1, 0 otherwise, and v is a column vectors with entries 1 for configurations where the average of no top row entry and its neighbours is 1, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](81, i -> `if`(Configs[i][3]+Configs[i][1]+Configs
  [i][4]<>3 and Configs[i][4]+Configs[i][2]+Configs[i][3]<>3,1,0)):

  v:= Vector[column](81, i -> `if`(Configs[i][1]+Configs[i][2]+
  Configs[i][3]<>3 and Configs[i][1]+Configs[i][2]+Configs[i][4]
  <>3,1,0)):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 50 do TV[n]:= T . TV[n-1] od:
> A:= [seq(u . TV[n],n=0..18)];
```

$$A := [36, 174, 922, 5202, 29180, 163706, 919924, 5171420, 29068726, 163403454, 918546962, 5163485576, 29025793202, 163164480016, 917206531060, 5155949803838, 28983459159770, 162926511803678, 915868877073788] \quad \textbf{(1)}$$

Note that $a(1) = 6$ is not included here. The verification of the empirical recurrence for $n = 15$, which involves $a(1)$, must be done separately.

```
> n:= 'n':
  empirical:= a(n) = 5*a(n-1) +8*a(n-2) -20*a(n-3) -23*a(n-4) -52*a
  (n-5) +25*a(n-6) +254*a(n-7) +62*a(n-8) -134*a(n-9) -88*a(n-10)
  -360*a(n-11) +176*a(n-12) -168*a(n-13) +184*a(n-14):
> eval(empirical, {a(n-14)=6, seq(a(n-i)=A[14-i],i=0..13)});
```

(2)

$$163164480016 = 163164480016 \tag{2}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto 376832\, t^3 + 32768\, t^4 + 15\, t^{63} - 22\, t^{62} + 29\, t^{61} + 340\, t^{60} - 592\, t^{59} + 858\, t^{58} - 2543\, t^{57} - 6012\, t^{56} + 8052\, t^{55} - 14014\, t^{54} + 51095\, t^{53} + 66984\, t^{52} - 43878\, t^{51} + 164390\, t^{50} - 526889\, t^{49} - 480692\, t^{48} - 83072\, t^{47} - 1497746\, t^{46} + 3262391\, t^{45} + 1892772\, t^{44} + 2740608\, t^{43} + 9470978\, t^{42} - 12642768\, t^{41} - 1503076\, t^{40} - 18412329\, t^{39} - 38593148\, t^{38} + 30785908\, t^{37} - 21733964\, t^{36} + 68351632\, t^{35} + 96781900\, t^{34} - 45560032\, t^{33} + 115181388\, t^{32} - 163270932\, t^{31} - 134852144\, t^{30} + 33375744\, t^{29} - 292189576\, t^{28} + 273330720\, t^{27} + 55870400\, t^{26} + 1702976\, t^{25} + 416442528\, t^{24} - 354793344\, t^{23} + 127756352\, t^{22} + 2708096\, t^{21} - 289894080\, t^{20} + 357555840\, t^{19} - 252298752\, t^{18} - 50634240\, t^{17} + 39003904\, t^{16} - 229675776\, t^{15} + 206082048\, t^{14} + 20889600\, t^{13} + 13802496\, t^{12} + 89369600\, t^{11} - 60088320\, t^{10} - 6193152\, t^9 - 102400\, t^8 - 11096064\, t^7 + 2670592\, t^6 + 393216\, t^5 + t^{65} - 8\, t^{64} \tag{3}$$

```
> degree(P(t));
```

$$65 \tag{4}$$

This turns out to have degree 65, but with the t^0, t^1 and t^2 coefficients 0. Thus we will have $0 = u\, P(T)\, T^n\, v = \sum_{i=3}^{65} p_i\, b(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 62, which would hold true for any u and v, after a delay of 2. It seems that with our particular u and v we have a recurrence of order only 14, corresponding to a factor of P .

```
> Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(14-i),i=0..14),t);
```

$$Q := t \mapsto t^{14} - 5\, t^{13} - 8\, t^{12} + 20\, t^{11} + 23\, t^{10} + 52\, t^9 - 25\, t^8 - 254\, t^7 - 62\, t^6 + 134\, t^5 + 88\, t^4 + 360\, t^3 - 176\, t^2 + 168\, t - 184 \tag{5}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 51, again with the lowest three coefficients 0.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(-2048 - 2048\, t - 18432\, t^3 - 2048\, t^2 + 40448\, t^4 + t^{48} - 3\, t^{47} + 8\, t^{46} - 26\, t^{45} - 11\, t^{43} - 130\, t^{42} + 481\, t^{41} - 106\, t^{40} + 1221\, t^{39} + 1457\, t^{38} - 2199\, t^{37} + 2754\, t^{36} - 11514\, t^{35} - 8045\, t^{34} - 2875\, t^{33} - 27565\, t^{32} + 43758\, t^{31} + 8621\, t^{30} + 46462\, t^{29} + 120577\, t^{28} - 86130\, t^{27} + 77434\, t^{26} - 152474\, t^{25} - 270860\, t^{24} + 98576\, t^{23} - 366480\, t^{22} + 259012\, t^{21} + 284484\, t^{20} - 122256\, t^{19} + 738400\, t^{18} - 305552\, t^{17} - 37920\, t^{16} + 269088\, t^{15} - 762784\, t^{14} + 388192\, t^{13} - 114080\, t^{12} - 463488\, t^{11} + 314752\, t^{10} - 403072\, t^9 - 59264\, t^8 + 353792\, t^7 + 1536\, t^6 + 48640\, t^5\right) t^3 \tag{6}$$

```
> degree(R(t));
```

$$51 \tag{7}$$

Now we want to show that $c(n) = u\, Q(T)\, T^n\, v = 0$ for all $n \geq 0$. This will certainly satisfy the recurrence

$$\sum_{i=3}^{51} r_i\, c(i+n) = \sum_{i=2}^{51} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(1) = \ldots = c(50) = 0$.

First we compute $w = u\, Q(T)$, then multiply it with the already-computed $T^n v$.

```
> UT[0]:= u:
  for n from 1 to 14 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..degree(Q(t))):
> seq(w . TV[n],n=0..50);
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 **(8)**

This completes the proof.